


Contact Elements

Introduction

This tutorial was completed using ANSYS 7.0 The purpose of the tutorial is to describe how to utilize contact elements to simulate how two beams react when they come into contact with each other.

The beams, as shown below, are 100mm long, 10mm x 10mm in cross-section, have a Young's modulus of 200 GPa, and are rigidly constrained at the outer ends. A 10KN load is applied to the center of the upper, causing it to bend and contact the lower.



Preprocessing: Defining the Problem

1. Give example a Title

Utility Menu > File > Change Title ...
/title, Contact Elements

2. Open preprocessor menu

ANSYS Main Menu > Preprocessor
/PREP7

3. Define Areas

Preprocessor > Modeling > Create > Area > Rectangle > By 2 Corners
BLC4,WP X, WP Y, Width, Height

We are going to define 2 rectangles as described in the following table:


Rectangle	Variables (WP X,WP Y,Width,Height)
1	(0, 15, 100, 10)
2	(50, 0, 100, 10)

4. Define the Type of Element

- Preprocessor > Element Type > Add/Edit/Delete...

For this problem we will use the PLANE42 (Solid, Quad 4node 42) element. This element has 2 degrees of freedom at each node (translation along the X and Y).

- While the Element Types window is still open, click **Options...** Change **Element behavior K3** to `Plane strs w/thk` as shown below. This allows a thickness to be input for the elements.



5. Define Real Constants

Preprocessor > Real Constants... > Add...

In the 'Real Constants for PLANE42' window, enter the following geometric properties:

- i. Thickness THK: 10

This defines a beam with a thickness of 10 mm.

6. Define Element Material Properties

Preprocessor > Material Props > Material Models > Structural > Linear > Elastic > Isotropic

In the window that appears, enter the following geometric properties for steel:

- . Young's modulus EX: 200000
- i. Poisson's Ratio PRXY: 0.3

7. Define Mesh Size

Preprocessor > Meshing > Size Cntrls > ManualSize > Areas > All Lines...

For this example we will use an element edge length of 2mm.

8. Mesh the frame


Preprocessor > Meshing > Mesh > Areas > Free > click 'Pick All'

9. Define the Type of Contact Element

- o Preprocessor > Element Type > Add/Edit/Delete...

For this problem we will use the CONTAC48 (Contact, pt-to-surf 48) element. CONTAC48 may be used to represent contact and sliding between two surfaces (or between a node and a surface) in 2-D. The element has two degrees of freedom at each node: translations in the nodal x and y directions. Contact occurs when the contact node penetrates the target line.

- o While the Element Types window is still open, click **Options...** Change **Contact time/load prediction K7** to Reasonabl T/L inc. This is an important step. It initiates a process during the solution calculations where the time step or load step, depending on what the user has specified in the solution controls, increments slowly when contact is imminent. This way, one surface won't penetrate too far into the other and cause the solution to fail.



It is important to note, CONTAC48 elements are created in the space between two surfaces prescribed by the user. This will be covered below. As the surfaces approach each other, the contact element is slowly "crushed" until its upper node(s) lie along the same line as the lower node(s). Thus, ANSYS can calculate when the two prescribed surfaces have made contact. Other contact elements, such as CONTA175, require a target element, such as TARGE169, to function. When using contact elements in your own analyses, be sure to understand how the elements work. The ANSYS help file has plenty of useful information regarding contact elements and is worth reading.

10. Define Real Constants for the Contact Elements

Preprocessor > Real Constants... > Add...

In the 'Real Constants for CONTAC48' window, enter the following properties:

- . Normal contact stiffness KN: 200000

CONTAC48 elements basically use a penalty approach to model contact. When one surface comes into "contact" with the other, ANSYS numerically puts a spring of stiffness KN between the two. ANSYS recommends a value between 0.01 and 100 times Young's modulus for the material. Since this "spring" is so stiff, the behaviour of the model is like the two surfaces have made contact. This KN value can greatly affect your solution, so be sure to read the help file on contact so you can recognize when your solution is not converging and why. A good rule of thumb is to start with a low value of KN and see how the solution converges (start watching the ANSYS Output Window). If there is too much penetration, you should increase KN. If it takes a lot of iterations to converge for a single substep, you should decrease KN.

- i. Target length tolerance TOLS: 10

Real constant TOLS is used to add a small tolerance that will internally increase the length of the target. This is useful for problems when node to node contact is likely to occur, rather than node to element edge. In this situation, the contact node may repeatedly "slip" off one of the target nodes, resulting in convergence difficulties. A small value of TOLS, given in %, is usually enough to prevent such difficulties.

The other real constants can be used to model sliding friction, tolerances, etc. Information about these other constants can be found in the help file.

11. Define Nodes for Creating Contact Elements

Unlike the normal meshing sequence used for most elements, contact elements must be defined in a slightly different manner. Sets of nodes that are likely to come into contact must be defined and used to generate the necessary elements. ANSYS has many recommendations about which nodes to select and whether they should act as target nodes or source nodes. In this simple case, source nodes are those that will move into contact with the other surface, where as target nodes are those that are contacted. These terms are

important when using the automatic contact element mesher to ensure the elements will correctly model contact between the surfaces. A strong understanding of how the elements work is important when using contact elements for your own analysis.

First, the source nodes will be selected.


- Utility Menu > Select > Entities...

Select **Areas** and **By Num/Pick** from the pull down menus, select **From Full** from the radio buttons and click OK. Select the top beam and click OK. This will ensure any nodes that are selected in the next few steps will be from the upper beam. In this case, it is not too hard to ensure you select the correct nodes. However, when the geometry is complex, you may inadvertently select a node from the wrong surface and it could cause problems during element generation.




- Utility Menu > Select > Entities...

Select **Nodes** and **By Location** from the pull down menus, **Y coordinates** and **Reselect** from the radio buttons and enter a value of 15 and click OK. This will select all nodes along the bottom of the upper beam.



- Utility Menu > Select > Entities...
Select **Nodes** and **By Location** from the pull down menus, **X coordinates** and **Reselect** from the radio buttons and enter values of 50,100. This will select the nodes above the lower beam.



- Now if you list the selected nodes, **Utility Menu > List > Nodes...** you should only have the following nodes remaining.

5


LIST ALL SELECTED NODES. DSYS= 0
SORT TABLE ON NODE NODE NODE

NODE	X	Y	Z	THXY	THYZ	THZX
307	50.000	15.000	0.0000	0.00	0.00	0.00
309	52.000	15.000	0.0000	0.00	0.00	0.00
310	54.000	15.000	0.0000	0.00	0.00	0.00
311	56.000	15.000	0.0000	0.00	0.00	0.00
312	58.000	15.000	0.0000	0.00	0.00	0.00
313	60.000	15.000	0.0000	0.00	0.00	0.00
314	62.000	15.000	0.0000	0.00	0.00	0.00
315	64.000	15.000	0.0000	0.00	0.00	0.00
316	66.000	15.000	0.0000	0.00	0.00	0.00
317	68.000	15.000	0.0000	0.00	0.00	0.00
318	70.000	15.000	0.0000	0.00	0.00	0.00
319	72.000	15.000	0.0000	0.00	0.00	0.00
320	74.000	15.000	0.0000	0.00	0.00	0.00
321	76.000	15.000	0.0000	0.00	0.00	0.00
322	78.000	15.000	0.0000	0.00	0.00	0.00
323	80.000	15.000	0.0000	0.00	0.00	0.00
324	82.000	15.000	0.0000	0.00	0.00	0.00
325	84.000	15.000	0.0000	0.00	0.00	0.00
326	86.000	15.000	0.0000	0.00	0.00	0.00
327	88.000	15.000	0.0000	0.00	0.00	0.00
NODE	X	Y	Z	THXY	THYZ	THZX
328	90.000	15.000	0.0000	0.00	0.00	0.00
329	92.000	15.000	0.0000	0.00	0.00	0.00
330	94.000	15.000	0.0000	0.00	0.00	0.00
331	96.000	15.000	0.0000	0.00	0.00	0.00
332	98.000	15.000	0.0000	0.00	0.00	0.00
333	100.00	15.000	0.0000	0.00	0.00	0.00

It is important to try and limit the number of nodes you use to create contact elements. If you have a lot of contact elements, it takes a great deal of computational time to reach a solution. In this case, the only nodes that could make contact with the lower beam are those directly above it, thus those are the only nodes we will use to create the contact elements.

- Utility Menu > Select > Comp/Assembly > Create Component

Enter the component name *Source* as shown below, and click OK. Now we can use this component, *Source*, as a list of nodes to be used in other functions. This can be very useful in other applications as well.



Now select the target nodes. Using the same procedure as above, select the nodes on the lower beam directly under the upper beam. Be sure to reselect all nodes before starting to select others. This is done by opening the entity select menu, **Utility Menu > Select > Entities...**, clicking the **Also Select** radio button, and click the **Select All** button. These values will be the ones you'll use.

- Click the lower area for the area select.
- The Y coordinate is 10
- The X coordinates vary from 50 to 100.


When creating the component this time, enter the name **Target**.

IMPORTANT: Be sure to reselect all the nodes before continuing. This is done by opening the entity select menu, **Utility Menu > Select > Entities...**, clicking the **Also Select** radio button, and click the **Sele All** button.

12. Generate Contact Elements


Main Menu > Preprocessor > Modeling > Create > Elements > Elem Attributes

Fill the window in as shown below. This ensures ANSYS knows that you are dealing with the contact elements and the associated real constants.



Main Menu > Preprocessor > Modeling > Create > Elements > Surf / Contact > Node to Surf


The following window will pop up. Select the node set SOURCE from the first drop down menu (Ccomp) and TARGET from the second drop down menu (Tcomp). The rest of the selections remain unchanged.



At this point, your model should look like the following.



Unfortunately, the contact elements don't get plotted on the screen so it is sometimes difficult to tell they are there. If you wish, you can plot the elements (Utility Menu > Plot > Elements) and turn on element numbering (Utility Menu > PlotCtrls > Numbering > Elem/Attrib numbering > Element Type Numbers). If you zoom in on the contact areas, you can see little purple stars (Contact Nodes) and thin purple lines (Target Elements) numbered "2" which correspond to the contact elements, shown below.



The preprocessor stage is now complete.

Solution Phase: Assigning Loads and Solving

1. Define Analysis Type


Solution > Analysis Type > New Analysis > Static

ANTYPE, 0

2. Set Solution Controls

- o Select **Solution > Analysis Type > Sol'n Control...**

The following image will appear:




Ensure the following selections are made under the 'Basic' tab (as shown above)

- Ensure Automatic time stepping is on. Automatic time stepping allows ANSYS to determine appropriate sizes to break the load steps into. Decreasing the step size usually ensures better accuracy, however, this takes time. The Automatic Time Step feature will determine an appropriate balance. This feature also activates the ANSYS bisection feature which will allow recovery if convergence fails.
- Enter 100 as the number of substeps. This will set the initial substep to $1/100^{\text{th}}$ of the total load.
- Enter a maximum number of substeps of 1000. This stops the program if the solution does not converge after 1000 steps.
- Enter a minimum number of substeps of 20.
- Ensure all solution items are written to a results file.

Ensure the following selection is made under the 'Nonlinear' tab (as shown below)

- Ensure Maximum Number of Iterations is set to 100

**NOTE**

There are several options which have not been changed from their default values. For more information about these commands, type `help` followed by the command into the command line.

3. These solution control values are extremely important in determining if your analysis will succeed or fail. If you have too few substeps, the contact nodes may be driven through the target elements before ANSYS "realizes" it has happened. In this case the solution will resemble that of an analysis that didn't have contact elements defined at all. Therefore it is important to choose a relatively large number of substeps initially to ensure the model is defined properly. Once everything is working, you can reduce the number of substeps to optimize the computational time. Also, if the maximum number of substeps or iterations is left too low, ANSYS may stop the analysis before it has a chance to converge to a solution. Again, leave these relatively high at first.

4. **Apply Constraints**

Solution > Define Loads > Apply > Structural > Displacement > On Lines


Fix the left end of the upper beam and the right end of the lower beam (ie all DOF constrained)

5. **Apply Loads**

Solution > Define Loads > Apply > Structural > Force/Moment > On Nodes

Apply a load of -10000 in the FY direction to the center of the top surface of the upper beam. Note, this is a point load on a 2D surface. This type of loading should be avoided since it will cause a singularity. However, the displacement or stress near the load is not of interest in this analysis, thus we will use a point load for simplicity.

The applied loads and constraints should now appear as shown in the figure below.



6. Solve the System

Solution > Solve > Current LS
SOLVE

Postprocessing: Viewing the Results

1. Open postprocessor menu

ANSYS Main Menu > General Postproc
/POST1

2. Adjust Graphical Scaling

Utility Menu > PlotCtrls > Style > Displacement Scaling

Click the **1.0 (true scale)** radio button, then click ok. This is of huge importance! I lost many hours trying to figure out why the contact elements weren't working, when in fact it was just due to the displacement scaling to which ANSYS defaulted. If you leave the scaling as default, many times it will look like your contact nodes have gone through the target elements.


3. Show the Stress Distribution in the Beams

General Postproc > Plot Results > Contour Plot > Nodal Solu > Stress > von Mises


4. Adjust Contour Scale

Utility Menu > PlotCtrls > Style > Contours > Non-Uniform Contours

Fill in the window as follows:



This should produce the following stress distribution plot:



As seen in the figure, the load on the upper beam caused it to deflect and come in contact with the lower beam, producing a stress distribution in both.

Command File Mode of Solution

The above example was solved using a mixture of the Graphical User Interface (or GUI) and the command language interface of ANSYS. This problem has also been solved using the ANSYS command language interface that you may want to browse. Copy and paste following code into Notepad or a similar text editor and save it to your computer. Now go to 'File > Read input from...' and select the file.

```

finish
/clear

/title,Contact Elements
/prep7

! Top Beam
X1=0
Y1=15
L1=100
H1=10

! Bottom Beam
X2=50
Y2=0
L2=100
H2=10

! Create Geometry
blc4,X1,Y1,L1,H1
blc4,X2,Y2,L2,H2

! define element type

ET,1,plane42      ! element type 1
keyopt,1,3,3      ! plane stress w/thick
type,1            ! activate element type 1
R, 1, 10          ! thickness 0.01

! define material properties

MP,EX, 1, 200e3    ! Young's modulus
MP,NUXY,1, 0.3     ! Poisson's ratio

! meshing

esize,2           ! set meshing size
amesh,all         ! mesh area 1

ET,2,contac48     ! defines second element type - 2D contact elements
keyo,2,7,1        ! contact time/load prediction
r,2,200000,,,,,10
TYPE,2            ! activates or sets this element type
real,2            ! activates or sets the real constants

! define contact nodes and elements

! first the contact nodes
asel,s,area,,1    ! select top area
nsla,s,1          ! select the nodes within this area
nsl,r,loc,y,Y1    ! select bottom layer of nodes in this area
nsl,r,loc,x,X2,(X2+L2/2) ! select the nodes above the other beam
cm,source,node    ! call this group of nodes 'source'

! then the target nodes
allsel           ! relect everything
asel,s,area,,2   ! select bottom area
nsla,s,1         ! select nodes in this area
nsl,r,loc,y,H2   ! select bottom layer of nodes in this area
nsl,r,loc,x,X2,(X2+L2/2) ! select the nodes above the other beam
cm,target,node   ! call this selection 'target'

gcgen,source,target,3 ! generate contact elements between defined nodes

```



```
finish
/solut
antype,0

time,1                ! Sets time at end of run to 1 sec
autots,on             ! Auto time-stepping on
nsubst,100,1000,20   ! Number of sub-steps
outres,all,all       ! Write all output
neqit,100            ! Max number of iterations

nselect,s,loc,x,X1    ! Constrain top beam
nselect,r,loc,y,Y1,(Y1+H1)
d,all,all
nselect,all

nselect,s,loc,x,(X2+L2) ! Constrain bottom beam
nselect,r,loc,y,Y2,(Y2+H2)
d,all,all
nselect,all

nselect,s,loc,x,(L1/2+X1) ! Apply load
nselect,r,loc,y,(Y1+H1)
f,all,fy,-10000
nselect,all

solve
finish

/post1
/dscale,1,1
/CVAL,1,20,40,80,160,320,640,1280,2560
PLNSOL,S,EQV,0,1
```

Reference

<http://www.mece.ualberta.ca/tutorials/ansys>